# Vakioiden variointi kolmannen kertaluvun yhtälölle

Olkoon tarkasteltavana kolmannen kertaluvun lineaarinen epähomogeeninen differentiaaliyhtälö

```
diffyht = (x - 1) y'''[x] - x y''[x] + y'[x] == Exp[x^2]
```

$y'[x] - x\,y''[x] + (-1 + x)\,y^{(3)}[x] == e^{x^2}$

Vastaava homogeeniyhtälö on

```
homogyht = (x - 1) y'''[x] - x y''[x] + y'[x] == 0
```

$y'[x] - x\,y''[x] + (-1 + x)\,y^{(3)}[x] == 0$

Tämän ratkaisut ovat suhteellisen yksinkertaiset ja pienellä pohdiskelulla arvattavissa. Jos *y*:n paikalle sijoitetaan eksponenttifunktio, niin yhtälö toteutuu:

```
homogyht /. y -> Exp // Simplify
```

```
True
```

Samoin käy, jos sijoitetaan funktio $x^2$. Tämä voidaan antaa Mathematicalle muodossa `#^2&`, missä `#` tarkoittaa funktion argumenttia ja & osoittaa funktiomäärittelyn päättymisen:

```
homogyht /. y -> (#^2 &) // Simplify
```

```
True
```

Kolmanneksi perusratkaisuksi sopii vakiofunktio, joka kaikkialla saa arvon 1. Seuraavassa on käytetty kolmatta erilaista tapaa määritellä funktio Mathematicalle:

```
homogyht /. y -> Function[x, 1] // Simplify
```

```
True
```

Homogeeniyhtälön perusjärjestelmä on siten

```
perusjarj = {Exp[x], x^2, 1}
```

$\{e^x, x^2, 1\}$

ja homogeeniyhtälön yleinen ratkaisu

```
homogrtk = perusjarj.{C[1], C[2], C[3]}
```

$e^x C[1] + x^2 C[2] + C[3]$

Epähomogeenisen yhtälön yksittäisratkaisu voidaan hakea vakioiden varioinnilla, jolloin yrite on

```
variot = {u[x], v[x], w[x]}
```

{u[x], v[x], w[x]}

```
yr0 = variot.perusjarj
```

$e^x u[x] + x^2 v[x] + w[x]$

Tämä sijoitetaan differentiaaliyhtälöön ja pyritään määrittämään sellaiset funktiot $u(x)$, $v(x)$ ja $w(x)$, että yhtälö toteutuu. Laskua yksinkertaistetaan kuten toisen kertaluvun yhtälön tapauksessakin asettamalla sopivia lisäehtoja. Toisen kertaluvun tapauksessa näitä on yksi, kolmannen kertaluvun tapauksessa kaksi.

Kerätään derivaatat listaksi:

```
derivaatat = D[variot, x]
```

{u′[x], v′[x], w′[x]}

Muodostetaan yritteen derivaatta

```
der1 = D[yr0, x]
```

$e^x u[x] + 2 x v[x] + e^x u'[x] + x^2 v'[x] + w'[x]$

ja yksinkertaistetaan sitä asettamalla funktioiden $u$, $v$ ja $w$ derivaattoja sisältävien termien summa nollaksi:

```
nollatermi1 = Coefficient[der1, derivaatat].derivaatat
```

$e^x u'[x] + x^2 v'[x] + w'[x]$

Tällöin derivaatta on

```
yr1 = der1 /. nollatermi1 -> 0
```

$e^x u[x] + 2 x v[x]$

Edellä on käytetty komentoja tarvittavan lisäehdon muodostamiseen ja derivaatan yksinkertaistamiseen. Tämä on luontevaa kirjoitettaessa ohjelmakoodia Mathematicalle, mutta interaktiivisessa laskennassa on yksinkertaisempaa poimia tarvittavat termit hiirellä.

Vastaavalla tavalla muodostetaan yritteen toinen derivaatta, asetetaan toinen lisäehto ja yksinkertaistetaan derivaattaa:

```
der2 = D[yr1, x]
```

$e^x u[x] + 2 v[x] + e^x u'[x] + 2 x v'[x]$

```
nollatermi2 = Coefficient[der2, derivaatat].derivaatat
```

$e^{x}\,u'[x]+2\,x\,v'[x]$

```
yr2 = der2 /. nollatermi2 -> 0
```

$e^{x}\,u[x]+2\,v[x]$

Yritteen kolmas derivaatta saadaan yksinkertaisesti derivoimalla; lisäehtoja ei enää aseteta:

```
yr3 = D[yr2, x]
```

$e^{x}\,u[x]+e^{x}\,u'[x]+2\,v'[x]$

Yritteen derivaatat sijoitetaan differentiaaliyhtälöön, jolloin saadaan vain funktioiden *u*, *v* ja *w* derivaattoja koskeva ehto. Funktiot itse supistuvat pois; tämä on seurausta siitä, että yrite muodostetaan homogeeniyhtälön ratkaisujen avulla.

```
ehto = diffyht /. {y[x] -> yr0, y'[x] -> yr1, y''[x] -> yr2, y'''[x] -> yr3} // Simplify
```

$(-1+x)\,(e^{x}\,u'[x]+2\,v'[x])==e^{x^2}$

Derivaatat saadaan ratkaistuiksi tästä ehdosta ja aiemmin asetetuista lisäehdoista:

```
derivrtk = Solve[{ehto, nollatermi1 == 0, nollatermi2 == 0}, derivaatat]
```

$$\left\{\left\{w'[x]\to\frac{e^{x^2}\,(-2+x)\,x}{2\,(-1+x)^2},\ u'[x]\to\frac{e^{-x+x^2}\,x}{(-1+x)^2},\ v'[x]\to-\frac{e^{x^2}}{2\,(-1+x)^2}\right\}\right\}$$

Funktiot *u*, *v* ja *w* saadaan tämän jälkeen integroimalla. Selkeintä on laskea funktiot määrättyinä integraaleina alarajan ollessa mielivaltainen. Mathematicassa voidaan aivan hyvin integroida listoja alkioittain:

```
integroitavat = derivaatat /. First[derivrtk] /. x -> t
```

$$\left\{\frac{e^{-t+t^2}\,t}{(-1+t)^2},\ -\frac{e^{t^2}}{2\,(-1+t)^2},\ \frac{e^{t^2}\,(-2+t)\,t}{2\,(-1+t)^2}\right\}$$

```
fktrtk = Integrate[integroitavat, {t, a, x}]
```

$$\left\{\int_a^x\frac{e^{-t+t^2}\,t}{(-1+t)^2}\,dt,\ \int_a^x-\frac{e^{t^2}}{2\,(-1+t)^2}\,dt,\ \int_a^x\frac{e^{t^2}\,(-2+t)\,t}{2\,(-1+t)^2}\,dt\right\}$$

Esiintyviä integraaleja ei onnistuta lausumaan alkeisfunktioiden tai Mathematican tuntemien funktioiden avulla.

Vakioiden varioinnilla on kuitenkin löydetty yksittäisratkaisu. Mathematican funktioksi määriteltynä tämä on

```
y0 = Function[x, Evaluate[fktrtk.perusjarj]]
```

$$\text{Function}\left[x,\ x^2\int_a^x-\frac{e^{t^2}}{2\,(-1+t)^2}\,dt+e^{x}\int_a^x\frac{e^{-t+t^2}\,t}{(-1+t)^2}\,dt+\int_a^x\frac{e^{t^2}\,(-2+t)\,t}{2\,(-1+t)^2}\,dt\right]$$

ja se toimii kuten mikä tahansa funktio:

**$y_0$[x]**

$$x^2 \int_a^x -\frac{e^{t^2}}{2\,(-1+t)^2}\,dt + e^x \int_a^x \frac{e^{-t+t^2}\,t}{(-1+t)^2}\,dt + \int_a^x \frac{e^{t^2}\,(-2+t)\,t}{2\,(-1+t)^2}\,dt$$

**$y_0$'[x]**

$$\frac{e^{x^2}\,x}{(-1+x)^2} + \frac{e^{x^2}\,(-2+x)\,x}{2\,(-1+x)^2} - \frac{e^{x^2}\,x^2}{2\,(-1+x)^2} + 2\,x \int_a^x -\frac{e^{t^2}}{2\,(-1+t)^2}\,dt + e^x \int_a^x \frac{e^{-t+t^2}\,t}{(-1+t)^2}\,dt$$

Ratkaisu voidaan sijoittaa differentiaaliyhtälöön ja tarkistaa, toteutuuko yhtälö:

**diffyht /. y -> $y_0$ // Simplify**

```
True
```

Differentiaaliyhtälön yleinen ratkaisu on siis

**homogrtk + $y_0$[x]**

$$e^x\,C[1] + x^2\,C[2] + C[3] + x^2 \int_a^x -\frac{e^{t^2}}{2\,(-1+t)^2}\,dt + e^x \int_a^x \frac{e^{-t+t^2}\,t}{(-1+t)^2}\,dt + \int_a^x \frac{e^{t^2}\,(-2+t)\,t}{2\,(-1+t)^2}\,dt$$

Nimittäjien nollakohdan takia tarkastelualue on joko $x < 1$ tai $x > 1$. Integraalin alaraja on valittava siitä alueesta, jota tarkastellaan.

Lukija tutkikoon, miten Mathematica ratkaisee edellä käsitellyn yhtälön suoraan `DSolve`-komennolla.

## Linkit

epähomogeenisen yhtälön ratkaisujoukko
vakioiden variointi toisen kertaluvun yhtälön tapauksessa
korkeampien kertalukujen lineaariyhtälöt

*SKK 27.04.2001*